



INSTITUTO UNIVERSITARIO AERONAUTICO

IMPLEMENTACION DE LA U.S. STANDARD ATMOSPHERE 1976

INFORME TECNICO DMA-38/2003

Autor:

Ing. Gustavo Scarpin

Revisión:

Ing. Víctor Torregiani
Jefe Dpto. Mec. Aer.

Vo.Bo. :

Brig. Ing. Fernando Álvarez
Decano Fac. de Ing.

Córdoba, 22 de Octubre de 2003



INDICE

	Pág.
RESUMEN	2
LISTA DE SIMBOLOS	3
1. INTRODUCCION	4
2. DESARROLLO	4
2.1 ALTITUD GEOPOTENCIAL, ALTITUD GEOMETRICA Y GRAVEDAD	4
2.2 ECUACIONES DE BASE	6
2.2.1 Ley de los gases perfectos	6
2.2.2 Ecuación hidrostática	6
2.2.3 Variación de la temperatura de escala molecular con la altitud	7
2.2.4 Variación del peso molecular medio con la altitud	7
2.3 PARAMETROS PRINCIPALES	8
2.3.1 Peso molecular	8
2.3.2 Temperatura de escala molecular	9
2.3.3 Presión	9
2.3.4 Densidad	9
2.4 CANTIDADES DERIVADAS	10
2.4.1 Aceleración de la gravedad	10
2.4.2 Peso específico	10
2.4.3 Densidad numérica	10
2.4.4 Altura de escala de presión	10
2.4.5 Velocidad media de las partículas de aire	10
2.4.6 Camino libre medio	11
2.4.7 Frecuencia de las colisiones	11
2.4.8 Velocidad del sonido	11
2.4.9 Coeficiente de viscosidad	11
2.4.10 Viscosidad cinemática	12
2.4.11 Coeficiente de conductividad térmica	12
2.4.12 Volumen molar	12
2.4.13 Presión adimensional	12
2.4.14 Temperatura adimensional	12
2.4.15 Densidad adimensional	12
2.5 CONSTANTES PRIMARIAS	13
3. REFERENCIAS	13
4. CONCLUSIONES	15
5. REFERENCIAS	15
ANEXO: SUBROUTINAS US1976.FOR Y DUS1976.FOR	16



IMPLEMENTACION DE LA U.S. STANDARD ATMOSPHERE 1976

Por:

Ing. Gustavo Scarpin

RESUMEN

En ingeniería aeronáutica tanto en el ámbito académico como laboral es común la necesidad de conocer los diferentes parámetros de la atmósfera estándar. Dada ésta necesidad y debido a que no existe una subrutina de fácil utilización se procedió a crear una en Fortran según las especificaciones de la U.S. Standard Atmosphere 1976¹ aplicable hasta los 86 km de altitud.

Por debajo de los 32 Km, esta atmósfera es idéntica a la Atmósfera Tipo de la OACI².

La U.S. Standard Atmosphere 1976 consiste de un modelo simple que representa una atmósfera idealizada, estacionaria, con moderada actividad solar. Los parámetros listados incluyen la temperatura, presión, densidad, aceleración causada por la gravedad, altura en la escala de presión, densidad numérica, velocidad media de las partículas de aire, peso molecular medio, velocidad del sonido, viscosidad dinámica, viscosidad cinemática, conductividad térmica y altitud geopotencial.

Las unidades están dadas en el sistema internacional SI.

En el Anexo se presenta la subrutina US1976.FOR y su manual de usuario.

Córdoba, 22 de octubre de 2003

**LISTA DE SIMBOLOS**

Símbolo	Unidad	Descripción
C_s	[m/s]	Velocidad del sonido
g	[m/s ²]	Aceleración de la gravedad
H	[m]	Altitud geopotencial
H_p	[m]	Altura en la escala de presión
k	[W/m 1/°K]	Conductividad térmica
L	[m]	Camino libre medio
L'_M	[°K/m]	Gradiente térmico de escala molecular, geopotencial.
M	[Kg/Kmol]	Peso molecular del aire
N_a	[1/Kmol]	Número de Avogadro
n	[1/m ³]	Densidad numérica
P	[Pa]	Presión
R^*	[J/(°K*Kmol)]	Constante universal de los gases
R_{Tierra}	[m]	Radio de la Tierra
S	[°K]	Constante de Sutherland
T	[°K]	Temperatura en las escalas termodinámicas absolutas
T_M	[°K]	Temperatura de escala molecular en las escalas termodinámicas absolutas
$\bar{V}$	[m/s]	Velocidad de partícula (media aritmética)
v	[m ³ /Kmol]	Volumen molar de aire bajo condiciones existentes de T y P
Z	[m]	Altitud geométrica
β	$[kg / (s \cdot m \cdot \sqrt{°K})]$	Constante usada en la ecuación de viscosidad de Sutherland
δ	[-]	Presión adimensional
γ	[-]	Relación de calores específicos
η	[m ² /s]	Viscosidad cinemática
θ	[-]	Temperatura adimensional
μ	[kg/m 1/s]	Coefficiente de viscosidad
ν	[1/s]	Frecuencia de colisión
ρ	[kg/m ³]	Densidad
σ	[-]	Densidad adimensional
Φ	[m ² /s ²]	Geopotencial (Energía potencial de la gravedad por unidad de masa)
ω	[N/m ³]	Peso específico
0		Subíndice que indica valor a nivel del mar

1. INTRODUCCION

El trabajo del U.S. Committee on Extension to the Standard Atmosphere (COESA), establecido en 1953, condujo a las versiones de la U.S. Standard Atmosphere de los años 1958, 1962, 1966 y 1976. Estos modelos fueron publicados en forma conjunta por el National Oceanic and Atmospheric Administration (NOAA), la National Aeronautics and Space Administration (NASA) y la U.S. Air Force. En el trabajo de la COESA participaron además 30 organizaciones de Estados Unidos representantes del gobierno, industria, institutos de desarrollo y universidades.

Basados en datos obtenidos por cohetes y satélites se determinó la variación de temperatura con la altura hasta los 1000 km. En este informe solamente se representa el segmento bajo de la atmósfera comprendida entre -5000 m a 86000 m.

A continuación se presenta el desarrollo de las ecuaciones usadas en la determinación de la atmósfera estándar.

2. DESARROLLO

2.1 ALTITUD GEOPOTENCIAL, ALTITUD GEOMETRICA Y GRAVEDAD

La altitud geométrica Z de un punto se define como la distancia, medida a lo largo de la línea de fuerza que pasa por el punto, desde la superficie equipotencial para la cual $\Phi=0$ hasta el punto en cuestión, la superficie para la cual $\Phi=0$ viene a coincidir con el nivel medio del mar. Las pequeñas diferencias entre la altitud geométrica, tal como aquí se define, y varias distancia en línea recta, indicadas en forma esquemática en la Figura 1, son despreciables para la mayoría de los fines prácticos en la gama de altitudes que se está considerando.




Figura 1: Relación entre varias alturas



Con esta definición, la relación diferencial entre la altitud geométrica Z y la geopotencial Φ es:

$$d\Phi = g \, dZ \quad (1)$$

que es el incremento de trabajo efectuado al levantar la unidad de masa a una distancia dZ venciendo la fuerza de gravedad. Por lo tanto, el geopotencial en un punto cuya altitud es Z viene dado por

$$\Phi = \int_0^Z g \, dZ \quad (2)$$

De acuerdo a la ecuación anterior que relaciona la altitud geométrica Z con el geopotencial Φ , la siguiente ecuación se emplea para definir la altitud geopotencial H :

$$H = \frac{1}{g_0} \Phi = \int_0^Z \frac{g}{g_0} \, dZ \quad (3)$$

en que g_0 denota el valor estándar a nivel del mar, de la aceleración debida a la gravedad.

Al igual que en el caso de la ec. (2), la integral en la ec. (3) se realiza a lo largo de la línea de fuerza que pasa por el punto en cuestión.

El significado de la altitud geométrica Z y de la altitud geopotencial H puede apreciarse en la Figura 1. Si se utiliza una plomada elemental para explorar la gravedad y si esta plomada gira con la tierra, entonces, empezando en el punto O y siguiendo hacia fuera, la línea vertical que proporciona la plomada marcará segmentos de línea que definirán la trayectoria curva OP . Bajo la influencia de fuerzas gravitatorias y centrífugas, esta línea OP se doblará hacia los polos a medida que se eleve, salvo a lo largo del eje de rotación de la Tierra y a lo largo de la prolongación de un radio ecuatorial. En la figura la longitud de arco ds medida a lo largo de la línea de fuerza de la gravedad de P a P' , es idéntica al incremento dZ en altitud geométrica según se ha definido. Es decir, la latitud geométrica Z es la distancia efectiva a lo largo de la línea OP .

Puesto que:

$$dH = \frac{g}{g_0} \, dZ \quad (4)$$

entonces, evidentemente, la altitud geopotencial se mide a lo largo de la línea OP pero difiere en valor numérico debido a la variación de la aceleración debida a la gravedad. Obsérvese que la escala en la Figura 1 se ha exagerado con el fin de mostrar la curva que engendra el hilo de la plomada en la altitud media (línea de fuerza), y también para mostrar claramente la relación entre la línea tangente, línea radial y las proyecciones.

Es importante tener en cuenta que la distancia medida en longitud física (como sería empleando una regla de un metro), entre superficies geopotenciales sucesivas que difieren en 1 unidad geopotencial, no es constante, sino que aumenta en distancia física a medida que aumenta la altura.

Para la atmósfera inferior en estudio es suficiente estimar la variación de la aceleración debida a la gravedad despreciando los efectos de la aceleración centrífuga por lo que resulta:



$$\frac{g}{g_0} = \left(\frac{R_{Tierra}}{Z + R_{Tierra}} \right)^2 \quad (5)$$

donde se adopta el radio de la tierra como $R_{Tierra} = 6356.766$ km.

Reemplazando la ec. (5) en (3) e integrando se obtiene:

$$H = \frac{R_{Tierra} Z}{R_{Tierra} + Z} \quad (6)$$

$$Z = \frac{R_{Tierra} H}{R_{Tierra} - H} \quad (7)$$

donde las alturas Z y H son virtualmente idénticas a baja alturas. Se puede observar que $Z=86000$ m corresponde a $H=84852$ m.

2.2 ECUACIONES DE BASE

Definir una atmósfera estándar es definir el estado del aire en todas las altitudes, es decir:

$$P = f(H) , \quad \rho = f(H) , \quad T_M = f(H) , \quad M = f(H) \quad (8)$$

por tal motivo hace falta cuatro ecuaciones que relacione estos parámetros, ellas son:

2.2.1 Ley de los gases perfectos

La ecuación de estado del gas perfecto (ley de los gases perfectos) es:

$$\rho = \frac{M P}{R^* T_M} \quad (9)$$

en que P es la presión atmosférica, ρ es la densidad del aire, R^* es la constante universal de los gases (8314.32 Joule/°K/Kmol), T la temperatura absoluta y M el peso molecular medio del aire.

2.2.2 Ecuación hidrostática

Al suponer que la atmósfera es estática respecto a la Tierra, la ecuación que relaciona la variación de presión con la altitud es:

$$dP = -\rho g dZ \quad (10)$$

siendo g la aceleración debida a la gravedad y Z la altitud geométrica.



2.2.3 Variación de la temperatura de escala molecular con la altitud

Se define la variación de la temperatura de escala molecular con la altitud geopotencial de la siguiente forma:

TABLA I : Variación de la temperatura T_M con H

Altitud Geopotencial H [m]	Temperatura T_M [°K]	Gradiente L'_M [°K/m]
0	288.15	-0.0065
11000	216.65	0.0
20000	216.65	0.001
32000	228.65	0.0028
47000	270.65	0.0
51000	270.65	-0.0028
71000	214.65	-0.002
84852	186.946	




FIGURA 1: Variación de la Temperatura en función de la altitud según atmósfera estándar

2.2.4 Variación del peso molecular medio con la altitud

La variación del peso molecular, respecto del peso molecular medio a nivel del mar ($M_0 = 28.9644$ [Kg/Kmol]) con la altitud geométrica se define de la siguiente manera:



TABLA II : Variación del peso molecular M con Z

Altitud Geométrica Z [m]	Relación de peso molecular M/M_0 [-]
0	1.000000
80000	1.000000
80500	0.999996
81000	0.999989
81500	0.999971
82000	0.999941
82500	0.999909
83000	0.999870
83500	0.999829
84000	0.999786
84500	0.999741
85000	0.999694
85500	0.999641
86000	0.999579




FIGURA 2: Variación del Peso Molecular con la altitud según atmósfera estándar

2.3 PARAMETROS PRINCIPALES

2.3.1. Peso molecular

El peso molecular medio a nivel del mar M_0 se supone igual a 28.9644 [Kg/Kmol] (escala: $C^{12}=12.0000$). La variación con la altitud geométrica está definida por la TABLA II.



2.3.2. Temperatura de escala molecular

La temperatura de escala molecular T_M se define:

$$T_M = \frac{M_0}{M} T \quad (11)$$

siendo T la temperatura en las escalas termodinámicas absolutas.

La temperatura de escala molecular se define como una serie de segmentos lineales conectados en altitud geopotencial hasta la altitud geométrica $Z = 86$ km. La forma general de cada segmento lineal es:

$$T_M = T_{M,b} + L'_M (H - H_b) + \Delta T_{isa} \quad (12)$$

donde L'_M es el gradiente de temperatura de escala molecular con altitud geopotencial $\frac{dT_M}{dH}$, H_b es la altitud geopotencial en la base de una determinada capa caracterizada por un valor específico de L'_M , $T_{M,b}$ es el valor de T_M a la altitud H_b , mientras que ΔT_{isa} es la desviación de la temperatura respecto de la estándar. Los valores básicos definidores y los gradientes figuran en la Tabla I.

2.3.3. Presión

Dentro de una capa atmosférica en toda la cual T_M es una función lineal de H , la ecuación hidrostática y la ley de los gases perfectos permiten establecer la siguiente expresión para la presión:

$$\ln P = \ln P_b - \frac{g_0}{L'_M} \frac{M_0}{R^*} \ln \frac{L'_M (H - H_b) + T_{M,b}}{T_{M,b}} \quad (L'_M \neq 0) \quad (13-a)$$

$$\ln P = \ln P_b - \frac{g_0 M_0}{R^*} \frac{1}{T_{M,b}} (H - H_b) \quad (L'_M = 0) \quad (13-b)$$

o bien

$$\frac{P}{P_b} = \left(\frac{T_{M,b}}{L'_M (H - H_b) + T_{M,b}} \right)^{\frac{g_0 M_0}{R^* L'_M}} \quad (L'_M \neq 0) \quad (14-a)$$

$$\frac{P}{P_b} = e^{\left(\frac{-g_0 M_0 (H - H_b)}{R^* T_{M,b}} \right)} \quad (L'_M = 0) \quad (14-b)$$

Estas expresiones, que están en función de la altitud geopotencial H , se han empleado para calcular las tablas.

2.3.4. Densidad

La densidad puede calcularse partiendo de la presión (§2.3.3), de la temperatura de escala molecular (§2.3.2) y del peso molecular (§2.3.1) por medio de la ley de los gases perfectos (ec. 9).



2.4 CANTIDADES DERIVADAS

2.4.1 Aceleración de la gravedad

Se define la aceleración de la gravedad a nivel del mar a la correspondiente a una latitud de 45°, siendo $g_0 = 9.80665 \text{ [m/s}^2\text{]}$. La variación de g con la altura geométrica está definida por la ec. (5).

2.4.2 Peso específico

El peso específico ω es el peso por unidad de volumen, es decir:

$$\omega = \rho g \quad (15)$$

Con la ec. (15) y la ley de los gases perfectos dan:

$$\omega = g \frac{R^*}{M_0} \frac{P}{T_M} \quad (16)$$

2.4.3 Densidad numérica

La densidad numérica del aire se define como el número de partículas atmosféricas por unidad de volumen, considerándose las partículas en estado neutro. La expresión es:

$$n = \frac{M_0}{R^*} \frac{N_a P}{MT_M} \quad (17)$$

en que N_a es el número de Avogadro ($N=6.022169E26 \text{ [1/Kmol]}$ constante).

2.4.4 Altura de escala de presión

La altura de presión es la altura en la atmósfera donde la presión es 1/e veces el valor en la superficie, es decir:

$$H_p = \frac{R^*}{M_0} \frac{T_M}{g} \quad (18)$$

Este parámetro es muy usado en el estudio atmosférico. Por ejemplo, para la temperatura $T_M=250 \text{ [}^\circ\text{K]}$, da $H_p = 7.4 \text{ km}$.

2.4.5 Velocidad media de las partículas de aire

La velocidad de las partículas de aire es la media aritmética de las velocidades de todas las partículas de aire en el elemento de volumen que se considere. Viene dada por la expresión:

$$\bar{V} = \sqrt{\frac{8}{\pi} \frac{R^*}{M_0} T_M} \quad (19)$$



2.4.6 Camino libre medio

El camino libre medio L es el valor medio de las distancias recorridas por cada una de las partículas neutras en un volumen seleccionado, entre colisiones sucesivas con otras partículas del mismo volumen. La fórmula para calcular L es:

$$L = \frac{1}{\sqrt{2} \pi N \sigma^2} \frac{R^*}{M_0} \frac{M T_M}{P} \quad (20)$$

donde σ es el diámetro efectivo de colisión de la partícula media de aire.

2.4.7 Frecuencia de las colisiones

La frecuencia media de las colisiones ν , es la velocidad media de las partículas de aire dentro de un volumen seleccionado dividida por el camino libre medio L de las partículas dentro de ese volumen. Es decir:

$$\nu = \frac{\bar{V}}{L} = 4\sigma^2 N_a \sqrt{\pi \frac{M_0}{R^*}} \frac{P}{M \sqrt{T_M}} \quad (21)$$

2.4.8 Velocidad del sonido

La expresión adoptada para la velocidad del sonido es

$$C_s = \sqrt{\gamma \frac{R^*}{M} T_M} \quad (22)$$

en que γ es la relación entre los calores específicos del aire a presión constante y a volumen constante, que se toma igual a 1.4 exactamente (adimensional)

La ec. (22) solamente se aplica cuando la onda sonora es una pequeña perturbación de las condiciones ambiente. Los valores calculados de C_s se ha visto que varían ligeramente respecto a valores determinados experimentalmente. La atenuación que existe al nivel del mar respecto a las altas frecuencias afecta a frecuencia sucesivamente más bajas a medida que disminuye la presión atmosférica o a medida que aumenta el camino libre medio. Por este motivo el concepto de velocidad del sonido (excepto por lo que se refiere a frecuencias que se aproximan a cero) pierde progresivamente su gama de aplicabilidad a grandes altitudes.

2.4.9 Coeficiente de viscosidad

El coeficiente de viscosidad se define como un coeficiente de fricción interna que se produce cuando las regiones de fluido adyacente se desplazan, relativamente, a velocidades diferentes. La siguiente expresión, que proviene de la teoría cinética, pero con constantes deducidas experimentalmente, se emplea para el cálculo de las tablas:

$$\mu = \frac{\beta T^{1.5}}{T + S} \quad (23)$$



donde T es la temperatura termodinámica, β es una constante igual a $1.458E-6 \left[kg / (s \cdot m \cdot \sqrt{^\circ K}) \right]$ (exacta), y S es la constante de Sutherland, igual a $110.4 \text{ } ^\circ K$ (exacta).

2.4.10 Viscosidad cinemática

La viscosidad cinemática se define como la relación entre el coeficiente de viscosidad y la densidad de un fluido; es decir:

$$\eta = \frac{\mu}{\rho} \quad (24)$$

2.4.11 Coeficiente de conductividad térmica

Para el coeficiente de conductividad térmica se aplica la siguiente expresión empírica:

$$k = \frac{6.325 \cdot 10^{-7} T^{1.5}}{T + 245.4 \cdot 10^{-(12/T)}} \quad (25)$$

2.4.12 Volumen molar

Se define como el volumen por mol; es decir:

$$v = \frac{M}{\rho} = \frac{R^*}{M_0} \frac{M T_M}{P} \quad (26)$$

2.4.13 Presión adimensional

Se define como la presión dividido por la presión estándar a nivel del mar $P_0 = 101325 \text{ [Pa]}$:

$$\delta = \frac{P}{P_0} \quad (27)$$

2.4.14 Temperatura adimensional

Se define como la temperatura molecular dividida por la temperatura molecular estándar a nivel del mar $T_0 = 288.15 \text{ [} ^\circ K \text{]}$:

$$\theta = \frac{T_M}{T_0} \quad (28)$$

2.4.15 Densidad adimensional

Se define como la densidad dividida por la densidad estándar a nivel del mar $\rho_0 = 1.225 \text{ [kg/m}^3 \text{]}$:

$$\sigma = \frac{\delta}{\theta} = \frac{\rho}{\rho_0} \quad (29)$$



2.5 CONSTANTES PRIMARIAS

Para fines de cálculo es necesario establecer valores numéricos para varias constantes apropiadas a la atmósfera terrestre. En algunos casos el valor óptimo de la constante se conoce con mayor precisión que la necesaria en las tablas de atmosféricas y, por lo tanto, es apropiado redondearlo a un valor adecuado. La siguiente tabla da valores numéricos adoptados como exactos para los cálculos que aparecen en este documento.

TABLA III: Resumen de constantes primarias

Símbolo	Valor	Unidades SI
P_0	101325	[Pa]
ρ_0	1.2250	[kg/m ³]
T_0	288.15	[°K]
g_0	9.80665	[m/s ²]
S	110.4	[°K]
β	1.458E-6	[kg / (s m $\sqrt{^\circ K}$)]
γ	1.4	[-]
σ	3.65E-10	[m]
N_a	6.02257E+26	[1/Kmol]
R^*	8314.32	[J/(°K*Kmol)]

3. RESULTADOS

A continuación se presenta una tabla con valores de la atmósfera definida en el Capítulo 2 utilizando el programa del ANEXO.

TABLA IV-a: U.S. STANDARD ATMOSPHERE 1976

Z	H	T	T _M	θ	P	δ	ρ	σ
[m]	[m]	[°K]	[°K]	[-]	[Pa]	[-]	[kg/m ³]	[-]
-5000	-5003.9	320.676	320.676	1.11288	177761	1.75437	1.93112	1.57643
0	0	288.150	288.150	1	101325	1	1.2250	1
5000	4996.1	255.676	255.676	0.88730	54048.3	0.533415	0.736428	0.601166
10000	9984.3	223.252	223.252	0.77478	26499.9	0.261534	0.41351	0.33756
15000	14964.7	216.650	216.650	0.75187	12111.8	0.119534	0.194755	0.158984
20000	19937.3	216.650	216.650	0.75187	5529.31	5.457E-02	8.891E-02	7.258E-02
25000	24902.1	221.552	221.552	0.76888	2549.22	2.516E-02	4.008E-02	3.272E-02
30000	29859.1	226.509	226.509	0.78608	1197.03	1.181E-02	1.841E-02	1.503E-02
35000	34808.3	236.513	236.513	0.82080	574.594	5.671E-03	8.463E-03	6.909E-03
40000	39749.9	250.350	250.350	0.86882	287.144	2.834E-03	3.996E-03	3.262E-03
45000	44683.7	264.164	264.164	0.91676	149.101	1.472E-03	1.966E-03	1.605E-03
50000	49609.8	270.650	270.650	0.93927	79.7791	7.874E-04	1.027E-03	8.383E-04
55000	54528.2	260.771	260.771	0.90498	42.5251	4.197E-04	5.681E-04	4.638E-04
60000	59439.0	247.021	247.021	0.85727	21.9587	2.167E-04	3.097E-04	2.528E-04
65000	64342.1	233.292	233.292	0.80962	10.9297	1.079E-04	1.632E-04	1.332E-04
70000	69237.6	219.585	219.585	0.76205	5.22090	5.153E-05	8.283E-05	6.762E-05
75000	74125.4	208.399	208.399	0.72323	2.38814	2.357E-05	3.992E-05	3.259E-05
80000	79005.7	198.639	198.639	0.68936	1.05247	1.039E-05	1.846E-05	1.507E-05
86000	84852.0	186.867	186.946	0.64878	0.37338	3.685E-06	6.958E-06	5.680E-06



TABLA IV-b U.S. STANDARD ATMOSPHERE 1976

Z	H	g	ω	H _p	n	$\bar{V}$	v	L
[m]	[m]	[-]	[N/m ³]	[m]	[m ⁻³]	[m/s]	[s ⁻¹]	[m]
-5000	-5003.9	9.82210	18.9677	9371.82	4.015E+25	484.155	1.151E+10	4.207E-08
0	0	9.80665	12.0131	8434.52	2.547E+25	458.945	6.919E+09	6.633E-08
5000	4996.1	9.79124	7.21055	7495.72	1.531E+25	432.310	3.918E+09	1.103E-07
10000	9984.3	9.77587	4.04242	6555.45	8.598E+24	403.970	2.056E+09	1.965E-07
15000	14964.7	9.76053	1.90091	6371.58	4.050E+24	397.952	9.539E+08	4.172E-07
20000	19937.3	9.74523	0.86645	6381.59	1.849E+24	397.952	4.355E+08	9.139E-07
25000	24902.1	9.72997	0.39002	6536.22	8.335E+23	402.429	1.985E+08	2.027E-06
30000	29859.1	9.71474	0.17885	6692.94	3.828E+23	406.906	9.220E+07	4.413E-06
35000	34808.3	9.69955	8.209E-02	6999.49	1.760E+23	415.795	4.331E+07	9.600E-06
40000	39749.9	9.68439	3.870E-02	7420.57	8.308E+22	427.784	2.104E+07	2.033E-05
45000	44683.7	9.66927	1.901E-02	7842.29	4.088E+22	439.428	1.063E+07	4.132E-05
50000	49609.8	9.65418	9.914E-03	8047.39	2.135E+22	444.790	5.621E+06	7.912E-05
55000	54528.2	9.63913	5.476E-03	7765.75	1.181E+22	436.597	3.053E+06	1.430E-04
60000	59439.0	9.62411	2.980E-03	7367.76	6.439E+21	424.931	1.620E+06	2.624E-04
65000	64342.1	9.60913	1.568E-03	6969.12	3.394E+21	412.954	829498	4.978E-04
70000	69237.6	9.59419	7.947E-04	6569.86	1.722E+21	400.638	408414	9.810E-04
75000	74125.4	9.57928	3.824E-04	6244.90	8.301E+20	390.301	191765	2.035E-03
80000	79005.7	9.56440	1.765E-04	5961.67	3.838E+20	381.051	86563.8	4.402E-03
86000	84852.0	9.54956	7.849E-05	5678.00	1.710E+20	371.586	37601.5	9.882E-03

TABLA IV-c: U.S. STANDARD ATMOSPHERE 1976

Z	H	C _s	μ	μ/μ_0	η	η/η_0	k	k/k ₀	M
[m]	[m]	[m/s]	[kg/m/s]	[-]	[m ² /s]	[-]	[W/m ² /K]	[-]	[Kg/Kmol]
-5000	-5003.9	358.986	1.942E-05	1.085	1.006E-05	0.689	2.786E-02	1.099	28.9644
0	0	340.294	1.789E-05	1.000	1.461E-05	1.000	2.534E-02	1.000	28.9644
5000	4996.1	320.546	1.628E-05	0.910	2.211E-05	1.514	2.274E-02	0.898	28.9644
10000	9984.3	299.532	1.458E-05	0.815	3.525E-05	2.413	2.007E-02	0.792	28.9644
15000	14964.7	295.070	1.422E-05	0.794	7.299E-05	4.997	1.951E-02	0.770	28.9644
20000	19937.3	295.070	1.422E-05	0.794	1.599E-04	10.946	1.951E-02	0.770	28.9644
25000	24902.1	298.389	1.448E-05	0.809	3.613E-04	24.738	1.993E-02	0.786	28.9644
30000	29859.1	301.709	1.475E-05	0.824	8.013E-04	54.859	2.034E-02	0.803	28.9644
35000	34808.3	308.300	1.529E-05	0.854	1.806E-03	123.654	2.117E-02	0.836	28.9644
40000	39749.9	317.189	1.601E-05	0.895	4.007E-03	274.293	2.231E-02	0.880	28.9644
45000	44683.7	325.823	1.671E-05	0.934	8.500E-03	581.875	2.343E-02	0.925	28.9644
50000	49609.8	329.799	1.704E-05	0.952	1.659E-02	1.136E+03	2.395E-02	0.945	28.9644
55000	54528.2	323.724	1.654E-05	0.924	2.912E-02	1.993E+03	2.316E-02	0.914	28.9644
60000	59439.0	315.074	1.584E-05	0.885	5.114E-02	3.501E+03	2.204E-02	0.870	28.9644
65000	64342.1	306.193	1.512E-05	0.845	9.262E-02	6.341E+03	2.091E-02	0.825	28.9644
70000	69237.6	297.061	1.438E-05	0.803	0.173575	1.188E+04	1.976E-02	0.780	28.9644
75000	74125.4	289.396	1.376E-05	0.769	0.344653	2.359E+04	1.882E-02	0.743	28.9644
80000	79005.7	282.538	1.321E-05	0.738	0.715575	4.899E+04	1.798E-02	0.710	28.9644
86000	84852.0	275.520	1.264E-05	0.707	1.53824	1.053E+05	1.714E-02	0.676	28.9555



4. CONCLUSIONES

Se procedió a confeccionar un programa de cómputos en Fortran que permite obtener los parámetros de la U.S. STANDARD ATMOSPHERE 1976 desde una altitud geométrica de -5000 a 86000 metros.

Los resultados obtenidos fueron validados comparándolos con los presentados en Ref. [2] y Ref. [3] observándose en ambos casos una correspondencia biunívoca. Por tal motivo, la subrutina US1976.FOR y su versión de doble precisión DUS1976.FOR se consideran válidas para ser usadas tanto con fines académicos como en trabajos de ingeniería.

5. REFERENCIAS

- [1] "U.S. Standard Atmosphere", 1976, U.S. Government Printing Office, Washington, D.C., 1976.
- [2] "MANUAL DE LA ATMÓSFERA TIPO DE LA OACI – ampliada hasta 32 kilómetros (105000 pies)", Segunda Edición, Doc 7488/2, OACI, 1964.
- [3] Steven S. Pietrobon, "U.S. Standard Atmosphere, 1976 – Programa ussa1976.pas", Australia, 30-Dic-1999, <http://sworld.com.au/steven/space/atmosphere/>



ANEXO

SUBROUTINAS US1976.FOR Y DUS1976.FOR

1. DESCRIPCION

En este Anexo se presenta la subrutina US1976.FOR, y su versión para doble precisión DUS1976.FOR, desarrollada en lenguaje FORTRAN. Las mismas calculan los parámetros característicos de la US Standard Atmosphere 1976 desde la altitud geométrica de -5000 a 86000 metros.

El sistema de unidades utilizado en el programa es el sistema internacional SI.

2. ARGUMENTOS DE ENTRADA

Los argumentos de entrada son:

- Z : Altitud geométrica [metros]
- DTISA : Desviación de temperatura respecto de t_0 en la escala molecular [grados]

3. ARGUMENTOS DE SALIDA

Estos argumentos son:

- RHO : Densidad del aire [Kg/m^3]
- P : Presión [N/m^2]
- TM : Temperatura de escala molecular [$^{\circ}\text{K}$]
- SIGMA : Densidad adimensional [-]
- DELTA : Presión adimensional [-]
- THETA : Temperatura adimensional [-]
- VELSON : Velocidad del sonido [m/s]
- MU : Coeficiente de viscosidad [Kg/(m*s)]
- ETA : Viscosidad cinemática [1/s]
- K : Coeficiente de conductividad térmica [$\text{W/(m*}^{\circ}\text{K)}$]



- G : Aceleración de la gravedad [m/s^2]
- H : Altitud geopotencial [m]
- OMEGA : Peso específico [N/m^3]
- VELMED : Velocidad media de las partículas de aire [m/s]
- L : Camino libre medio [m]
- UN : Frecuencia de las colisiones [1/s]
- N : Densidad numérica [$1/\text{m}^3$]
- V : Volumen molar [m^3/Kmol]
- M : Peso molecular medio del aire [Kg/Kmol]
- HP : Altura en la escala de presión [m]
- T : Temperatura en las escalas termodinámicas absolutas [$^{\circ}\text{K}$]

4. LISTADO DE CÓDIGOS FUENTES

Primero se presenta el listado en código fuente de las subrutinas US1976.FOR y DUS1976.FOR. Finalmente se muestra a modo de ejemplo un programa que utiliza la subrutina US1976.FOR.

4.1 SUBROUTINA US1976.FOR (Precisión simple)

```
*****
**                                                                 **
**          SUBROUTINA US Standard Atmosphere 1976              **
**                                                                 **
*****
**                                                                 **
** - Cálculo de atmósfera estándar según el modelo US Standard  **
**   Atmosphere 1976.                                           **
**                                                                 **
** - Rango válido de altitudes: -5000 hasta 86000 metros.      **
**                                                                 **
** - Definida para una latitud de 45 grados.                    **
**                                                                 **
** - Unidades en el Sistema Internacional SI                     **
**                                                                 **
*****
**                                                                 **
** PARAMETROS DE ENTRADA:                                       **
**                                                                 **
** Z          : ALTITUD GEOMETRICA [METROS]                     **
** DTISA      : DESVIACION DE TEMPERATURA RESPECTO DE ISA EN LA **
**               ESCALA MOLECULAR [GRADOS]                      **
**                                                                 **
** PARAMETROS DE SALIDA:                                        **
**                                                                 **
** RHO        : DENSIDAD DEL AIRE [KG/M3]                       **
** P          : PRESION [N/M2]                                  **
** TM         : TEMPERATURA DE ESCALA MOLECULAR [°K]            **
** SIGMA      : DENSIDAD ADIMENSIONAL [-]                      **
**                                                                 **
```



```
** DELTA      : PRESION ADIMENSIONAL [-]                **
** THETA      : TEMPERATURA ADIMENSIONAL [-]            **
** VELSON     : VELOCIDAD DEL SONIDO [M/S]               **
** MU         : COEFICIENTE DE VISCOSIDAD [KG/(M*S)]     **
** ETA        : VISCOSIDAD CINEMATICA [1/S]             **
** K          : COEFICIENTE DE CONDUCTIVIDAD TERMICA [W/(M*°K)] **
** G          : ACELERACION DE LA GRAVEDAD [M/S2]        **
** H          : ALTITUD GEOPOTENCIAL [M]                 **
** OMEGA      : PESO ESPECIFICO [N/M3]                   **
** VELMED     : VELOCIDAD MEDIA DE LAS PARTICULAS DE AIRE [M/S] **
** L          : CAMINO LIBRE MEDIO [M]                   **
** NU         : FRECUENCIA DE LAS COLISIONES [1/S]       **
** N          : DENSIDAD NUMERICA [1/M3]                 **
** V          : VOLUMEN MOLAR [M3/KMOL]                  **
** M          : PESO MOLECULAR MEDIO DEL AIRE [KG/KMOL]  **
** HP         : ALTURA EN LA ESCALA DE PRESION [M]      **
** T          : TEMPERATURA EN LAS ESCALAS TERMODINAMICAS ABSOLUTAS [°K] **
**                                                    **
*****
**                                                    **
** AUTOR : Ing. Gustavo H. Scarpin                    ** CORDOBA, 20-OCT-2003 **
**                                                    **
*****
```

```
SUBROUTINE US1976(Z,DTISA,RHO,P,TM,SIGMA,DELTA,THETA,VELSON,MU,ETA
.
,K,G,H,OMEGA,VELMED,L,NU,N,V,M,HP,T)
```

```
IMPLICIT NONE
```

```
REAL Z,DTISA,RHO,P,T,SIGMA,DELTA,THETA,VELSON,MU,ETA,K,G,H,OMEGA,
.
VELMED,L,NU,N,V,RTIERRA,RHO0,G0,BETA,SUTH,GAMMA,NABO,R,M0,
.
TGRAD,TBASE,DELTAH,GMR,PI,SI,M,HP,TM,Zi
```

```
INTEGER NTAB1,NTAB2,i,j,ik
```

```
*** CONSTANTES LOCALES ***
PARAMETER (RTIERRA=6356766.0) !RADIO DE LA TIERRA [M]
PARAMETER (G0=9.80665) !ACELERACION DE LA GRAVEDAD A NIVEL DEL MAR [M/S]
PARAMETER (BETA=1.458E-6) !CONSTANTE DE VISCOSIDAD [KG/(S*M*SQRT(°K)]
PARAMETER (SUTH=110.4) !CONSTANTE DE SUTHERLAND [°K]
PARAMETER (GAMMA=1.4) !CONSTANTE ADIABATICA [-]
PARAMETER (NABO=6.02257E+26) !NUMERO DE ABOGADRO [1/(KMOL)]
PARAMETER (R=8314.32) !CONSTANTE UNIVERSAL DE LOS GASES [J/(°K*KMOL)]
PARAMETER (M0=28.9644) !PESO MOLECULAR [KG/KMOL]
PARAMETER (SI=3.65E-10) !DIAMETRO MEDIO DE COLISION [M]
PARAMETER (PI=3.141592654) !PI [-]
PARAMETER (NTAB1=8) !NUMERO DE ELEMENTOS DE LA TABLA1 DE DATOS
PARAMETER (NTAB2=13) !NUMERO DE ELEMENTOS DE LA TABLA2 DE DATOS
```

```
** NOTA: Se considera constante con la altitud a: M0 y SI.
```

```
*** ARREGLOS LOCALES ***
REAL HTAB(NTAB1),TTAB(NTAB1),PTAB(NTAB1),GTAB(NTAB1),MTAB(NTAB2)
```

```
** TABLA 1
DATA HTAB/ 0.0, 11000.0, 20000.0, 32000.0, 47000.0,
.
51000.0, 71000.0, 84852.0/

DATA TTAB/ 288.15, 216.65, 216.65, 228.65, 270.65,
```



```
.          270.65,    214.65,    186.946/

DATA PTAB/101325.0, 22632.06,  5474.889, 868.0187, 110.9063,
.          66.93887,  3.956420, 0.3733836/

DATA GTAB/ -0.0065,      0.0,      0.001,   0.0028,      0.0,
.          -0.0028,    -0.002,      0.0/

**      TABLA 2
DATA MTAB/1.000000, 0.999996,  0.999989, 0.999971, 0.999941,
.          0.999909, 0.999870,  0.999829, 0.999786, 0.999741,
.          0.999694, 0.999641,  0.999579/

***      BLOQUE DE CALCULO      ***

***      VERIFICACION DE ALTITUD MINIMA Y MAXIMA
IF (Z.LT.-5000.) STOP ' ALTURA MENOR QUE -5000 METROS'
IF (Z.GT.86000.) STOP ' ALTURA MAYOR QUE 86000 METROS'

****      PESO MOLECULAR
IF (Z.LT.80000.) THEN
    M=M0
ELSE
    Zi=(Z-80000.)/500.+1
    i=INT(Zi)
    M=M0*((MTAB(i+1)-MTAB(i))*(Zi-i)+MTAB(i))
END IF

****      ALTITUD GEOPOTENCIAL
H=Z*RTIERRA/(Z+RTIERRA)

****      BUSQUEDA DEL INTERVALO DE CALCULO
i=1
j=NTAB1

DO WHILE (j.GT.i+1)
    ik=(i+j)/2
    IF (H.LT.HTAB(ik)) THEN
        j=ik
    ELSE
        i=ik
    END IF
END DO

****      TEMPERATURA
TGRAD=GTAB(i)
TBASE=TTAB(i)
DELTAH=H-HTAB(i)
TM=TBASE+DTISA+TGRAD*DELTAH          !TEMPERATURA A ESCALA MOLECULAR
T=M/M0*TM
THETA=TM/TTAB(1)                    !TEMPERATURA ADIMENSIONAL

****      PRESION
GMR=G0*M0/R
IF (TGRAD.EQ.0.0) THEN                !PRESION
    P=PTAB(i)*EXP(-GMR*DELTAH/TBASE)
ELSE
    P=PTAB(i)*(TBASE/TM)**(GMR/TGRAD)
END IF
```



```
DELTA=P/PTAB (1)                                !PRESION ADIMENSIONAL

**** DENSIDAD
RHO0=M0*PTAB (1) / (R*TTAB (1) )                !DENSIDAD A NIVEL DEL MAR
SIGMA=DELTA/THETA                                !DENSIDAD ADIMENSIONAL
RHO=SIGMA*RHO0                                    !DENSIDAD

**** VELOCIDAD DEL SONIDO
VELSON=SQRT (GAMMA*R*TM/M0)

**** COEFICIENTE DE VISCOSIDAD
MU=BETA*T**1.5 / (T+SUTH)

**** VISCOSIDAD CINEMATICA
ETA=MU/RHO

**** COEFICIENTE DE CONDUCTIVIDAD TERMICA
K=6.325E-07*T** (1.5) / (T+245.4*10** (-12./T)) *3.6E6/860.

**** ACELERACION DE LA GRAVEDAD
G=G0*(RTIERRA/ (Z+RTIERRA) ) **2

**** PESO ESPECIFICO DEL AIRE
OMEGA=RHO*G

**** VELOCIDAD MEDIA DE LAS PARTICULAS DE AIRE
VELMED=SQRT (8*R*TM/ (PI*M0) )

**** CAMINO LIBRE MEDIO
L=R*TM*M/ (SQRT (2.) *M0*P*PI*NABO*SI**2)

**** FRECUENCIA DE COLISIONES
NU=VELMED/L

**** DENSIDAD NUMERICA
N=NABO*M0*P/ (R*M*TM)

**** VOLUMEN MOLAR
V=M/RHO

**** ALTURA DE ESCALA DE PRESION
HP=R*TM/ (M0*G)

RETURN

END SUBROUTINE US1976
```

4.2 SUBROUTINE DUS1976.FOR (Doble precisión)

```
*****
**
**              SUBROUTINE US Standard Atmosphere 1976
**
**              VERSION DOBLE PRECISION
*****
**
```



```
**      - Cálculo de atmósfera estándar según el modelo US Standard      **
**      Atmosphere 1976.                                                  **
**                                                                           **
**      - Rango válido de altitudes: -5000 hasta 86000 metros.          **
**                                                                           **
**      - Definida para una latitud de 45 grados.                        **
**                                                                           **
**      - Unidades en el Sistema Internacional SI                         **
**                                                                           **
*****
**
**      PARAMETROS DE ENTRADA:
**
**      Z          : ALTITUD GEOMETRICA [METROS]
**      DTISA      : DESVIACION DE TEMPERATURA RESPECTO DE ISA EN LA ESCALA
**                  MOLECULAR [GRADOS]
**
**      PARAMETROS DE SALIDA:
**
**      RHO        : DENSIDAD DEL AIRE [KG/M3]
**      P          : PRESION [N/M2]
**      TM         : TEMPERATURA DE ESCALA MOLECULAR [°K]
**      SIGMA      : DENSIDAD ADIMENSIONAL [-]
**      DELTA      : PRESION ADIMENSIONAL [-]
**      THETA      : TEMPERATURA ADIMENSIONAL [-]
**      VELSON     : VELOCIDAD DEL SONIDO [M/S]
**      MU         : COEFICIENTE DE VISCOSIDAD [KG/(M*S)]
**      ETA        : VISCOSIDAD CINEMATICA [1/S]
**      K          : COEFICIENTE DE CONDUCTIVIDAD TERMICA [W/(M*°K)]
**      G          : ACELERACION DE LA GRAVEDAD [M/S2]
**      H          : ALTITUD GEOPOTENCIAL [M]
**      OMEGA      : PESO ESPECIFICO [N/M3]
**      VELMED     : VELOCIDAD MEDIA DE LAS PARTICULAS DE AIRE [M/S]
**      L          : CAMINO LIBRE MEDIO [M]
**      NU         : FRECUENCIA DE LAS COLISIONES [1/S]
**      N          : DENSIDAD NUMERICA [1/M3]
**      V          : VOLUMEN MOLAR [M3/KMOL]
**      M          : PESO MOLECULAR MEDIO DEL AIRE [KG/KMOL]
**      HP         : ALTURA EN LA ESCALA DE PRESION [M]
**      T          : TEMPERATURA EN LAS ESCALAS TERMODINAMICAS ABSOLUTAS [°K]
**
*****
**
**      AUTOR : Ing. Gustavo H. Scarpin          ** CORDOBA, 20-OCT-2003 **
**
*****
```

```
SUBROUTINE DUS1976(Z,DTISA,RHO,P, TM, SIGMA, DELTA, THETA, VELSON, MU,
.                  ETA, K, G, H, OMEGA, VELMED, L, NU, N, V, M, HP, T)
```

```
IMPLICIT NONE
```

```
REAL*8 Z,DTISA,RHO,P,T,SIGMA,DELTA,THETA,VELSON,MU,ETA,K,G,H,OMEGA
.      ,VELMED,L,NU,N,V,RTIERRA,RHO0,G0,BETA,SUTH,GAMMA,NABO,R,M0,
.      ,TGRAD,TBASE,DELTAH,GMR,PI,SI,M,HP, TM, Zi
```

```
INTEGER NTAB1,NTAB2,i,j,ik
```

```
***      CONSTANTES LOCALES      ***
PARAMETER (RTIERRA=6356766.0) !RADIO DE LA TIERRA [M]
```



```
PARAMETER (G0=9.80665) !ACELERACION DE LA GRAVEDAD A NIVEL DEL MAR [M/S]
PARAMETER (BETA=1.458D-6) !CONSTANTE DE VISCOSIDAD [KG/(S*M*SQRT(°K)]
PARAMETER (SUTH=110.4) !CONSTANTE DE SUTHERLAND [°K]
PARAMETER (GAMMA=1.4) !CONSTANTE ADIABATICA [-]
PARAMETER (NABO=6.02257D+26) !NUMERO DE ABOGADRO [1/(KMOL)]
PARAMETER (R=8314.32) !CONSTANTE UNIVERSAL DE LOS GASES [J/(°K*KMOL)]
PARAMETER (M0=28.9644) !PESO MOLECULAR [KG/KMOL]
PARAMETER (SI=3.65D-10) !DIAMETRO MEDIO DE COLISION [M]
PARAMETER (PI=3.141592654) !PI [-]
PARAMETER (NTAB1=8) !NUMERO DE ELEMENTOS DE LA TABLA1 DE DATOS
PARAMETER (NTAB2=13) !NUMERO DE ELEMENTOS DE LA TABLA2 DE DATOS
```

** NOTA: Se considera constante con la altitud a: M0 y SI.

```
*** ARREGLOS LOCALES ***
REAL HTAB(NTAB1), TTAB(NTAB1), PTAB(NTAB1), GTAB(NTAB1), MTAB(NTAB2)
```

```
** TABLA 1
DATA HTAB/ 0.0, 11000.0, 20000.0, 32000.0, 47000.0,
. 51000.0, 71000.0, 84852.0/

DATA TTAB/ 288.15, 216.65, 216.65, 228.65, 270.65,
. 270.65, 214.65, 186.946/

DATA PTAB/101325.0, 22632.06, 5474.889, 868.0187, 110.9063,
. 66.93887, 3.956420, 0.3733836/

DATA GTAB/ -0.0065, 0.0, 0.001, 0.0028, 0.0,
. -0.0028, -0.002, 0.0/
```

```
** TABLA 2
DATA MTAB/1.000000, 0.999996, 0.999989, 0.999971, 0.999941,
. 0.999909, 0.999870, 0.999829, 0.999786, 0.999741,
. 0.999694, 0.999641, 0.999579/
```

```
*** BLOQUE DE CALCULO ***
```

```
*** VERIFICACION DE ALTITUD MINIMA Y MAXIMA
IF (Z.LT.-5000.D0) STOP ' ALTURA MENOR QUE -5000 METROS'
IF (Z.GT.86000.D0) STOP ' ALTURA MAYOR QUE 86000 METROS'
```

```
**** PESO MOLECULAR
IF (Z.LT.80000.D0) THEN
  M=M0
ELSE
  Zi=(Z-80000.D0)/500.D0+1.D0
  i=IDINT(Zi)
  M=M0*((MTAB(i+1)-MTAB(i))*(Zi-i)+MTAB(i))
END IF
```

```
**** ALTITUD GEOPOTENCIAL
H=Z*RTIERRA/(Z+RTIERRA)
```

```
**** BUSQUEDA DEL INTERVALO DE CALCULO
i=1
j=NTAB1

DO WHILE (j.GT.i+1)
```



```
        ik=(i+j)/2
        IF (H.LT.HTAB(ik)) THEN
            j=ik
        ELSE
            i=ik
        END IF
    END DO

****  TEMPERATURA
    TGRAD=GTAB(i)
    TBASE=TTAB(i)
    DELTAH=H-HTAB(i)
    TM=TBASE+DTISA+TGRAD*DELTAH           !TEMPERATURA A ESCALA MOLECULAR
    T=M/M0*TM
    THETA=TM/TTAB(1)                     !TEMPERATURA ADIMENSIONAL

****  PRESION
    GMR=G0*M0/R
    IF (TGRAD.EQ.0.0D0) THEN              !PRESION
        P=PTAB(i)*DEXP(-GMR*DELTAH/TBASE)
    ELSE
        P=PTAB(i)*(TBASE/TM)**(GMR/TGRAD)
    END IF
    DELTA=P/PTAB(1)                      !PRESION ADIMENSIONAL

****  DENSIDAD
    RHO0=M0*PTAB(1)/(R*TTAB(1))           !DENSIDAD A NIVEL DEL MAR
    SIGMA=DELTA/THETA                     !DENSIDAD ADIMENSIONAL
    RHO=SIGMA*RHO0                       !DENSIDAD

****  VELOCIDAD DEL SONIDO
    VELSON=DSQRT(GAMMA*R*TM/M0)

****  COEFICIENTE DE VISCOSIDAD
    MU=BETA*T**1.5D0/(T+SUTH)

****  VISCOSIDAD CINEMATICA
    ETA=MU/RHO

****  COEFICIENTE DE CONDUCTIVIDAD TERMICA
    K=6.325D-07*T**(1.5D0)/(T+245.4D0*10.D0**(-12.D0/T))*3.6D6/860.D0

****  ACELERACION DE LA GRAVEDAD
    G=G0*(RTIERRA/(Z+RTIERRA))**2.D0

****  PESO ESPECIFICO DEL AIRE
    OMEGA=RHO*G

****  VELOCIDAD MEDIA DE LAS PARTICULAS DE AIRE
    VELMED=DSQRT(8.D0*R*TM/(PI*M0))

****  CAMINO LIBRE MEDIO
    L=R*TM*M/(DSQRT(2.D0)*M0*P*PI*NABO*SI**2.D0)

****  FRECUENCIA DE COLISIONES
    NU=VELMED/L

****  DENSIDAD NUMERICA
    N=NABO*M0*P/(R*M*TM)
```



```
**** VOLUMEN MOLAR
      V=M/RHO

**** ALTURA DE ESCALA DE PRESION
      HP=R*TM/ (M0*G)

      RETURN

      END SUBROUTINE DUS1976
```

4.3 PROGRAMA ZUS1976.FOR (Precisión simple)

Programa de ejemplo de utilización de la subrutina US1976.FOR.

```
***  PROGRAMA COMO EJEMPLO DE USO DE LA SUBRRUTINA US1976.FOR

      PROGRAM ZUS1976

      REAL Z,DTISA,RHO,P,T,SIGMA,DELTA,THETA,VELSON,MU,ETA,
      .      K,G,H,OMEGA,VELMED,L,NU,N,V,M,HP,TM

      NAMELIST/SALIDA/Z,DTISA,RHO,P,T,SIGMA,DELTA,THETA,VELSON,MU,ETA,
      .      K,G,H,OMEGA,VELMED,L,NU,N,V,M,HP,TM

      OPEN(UNIT=3,FILE='US1976.OUT',STATUS='UNKNOWN')
      OPEN(UNIT=4,FILE='TABLA.OUT',STATUS='UNKNOWN')

      DTISA=0.0

      DO Z=-5000.,86000.,5000.

          CALL US1976(Z,DTISA,RHO,P,TM,SIGMA,DELTA,THETA,VELSON,MU,ETA,
      .      K,G,H,OMEGA,VELMED,L,NU,N,V,M,HP,T)

          WRITE(3,SALIDA)
          WRITE(*,SALIDA)

          WRITE(4,'(30G15.6)') Z,H,T,TM,THETA,P,DELTA,RHO,SIGMA,G,N,VELMED
      .      ,NU,L,VELSON,MU,ETA,K,M,HP,OMEGA

      END DO

      END
```